状态是一维的。

我用这题的思想来为大家入个门。

下降/非降子序列问题：

问题描述： {挖掘题目的本质，一但抽象成这样的描述就可以用这个方法解}
在一个无序的序列a1,a2,a3,a4…an里，找到一个最长的序列满足：ai<=aj<=ak…<=am,且i<j<k…<m.（最长非降子序列） 或ai>aj>ak…>am,且i>j>k…>m.（最长下降子序列）。

PS：如果前i-1个数中用到ak (ak>ai或ak<=ai)构成了一个的最长的序列加上第I个数ai就是前i个数中用到i的最长的序列了。那么求用到ak构成的最长的序列有要求前k-1个数中……

从上面的分析可以看出这样划分问题满足最优子结构，那满足无后效性么？显然对于第i个数时只考虑前i-1个数,显然满足无后效性，可以用动态规划解。

分析到这里动态规划的三要素就不难得出了：如果按照序列编号划分阶段，设计一个状态opt[i] 表示前i个数中用到第i个数所构成的最优解。那么决策就是在前i-1个状态中找到最大的opt[j]使得aj>ai(或aj<=ai)，opt[j]+1就是opt[i]的值；用方程表示为：

opt[i]=max(opt[j])+1   (0<=j<i 且aj<=ai)       {最长非降子序列}
opt[i]=max(opt[j])+1   (0<=j<i 且aj>ai)        {最长下降子序列}
实现求解的部分代码：

      opt[0]:=maxsize；{maxsize 为maxlongint或-maxlongint}
for i:=1 to n do（这里是从头到尾，当然也可以从尾到头，可以这样表示 for i:=n downto1 do）

for j:=0 to i-1 do(上面如果是从尾到头，那么这里可以for j:=i+1 to n do)
if ( a[j]>a[i]) and (opt[j]+1>opt[i]) then       
opt[i]:=opt[j]+1;
ans:=-maxlongint;
for i:=1 to n do 
if opt[i]>ans then ans:=opt[i];                {ans 为最终解} 
复杂度：从上面的实现不难看出时间复杂度为O（N2），空间复杂度O（N）；

拦截导弹
(missile.pas/c/cpp)
来源：NOIP1999(提高组) 第一题

【问题描述】

    某国为了防御敌国的导弹袭击，发展出一种导弹拦截系统。但是这种导弹拦截系统有一个缺陷：虽然它的第一发炮弹能够到达任意的高度，但是以后每一发炮弹都不能高于前一发的高度。某天，雷达捕捉到敌国的导弹来袭。由于该系统还在试用阶段，所以只有一套系统，因此有可能不能拦截所有的导弹。

   输入导弹依次飞来的高度（雷达给出的高度数据是不大于30000的正整数），计算这套系统最多能拦截多少导弹，如果要拦截所有导弹最少要配备多少套这种导弹拦截系统。

【输入文件】missile.in
单独一行列出导弹依次飞来的高度。

【输出文件】missile.out
两行，分别是最多能拦截的导弹数，要拦截所有导弹最少要配备的系统数

【输入样例】

389 207 155 300 299 170 158 65
【输出样例】

6
2
PS：这道题的第一问我们不难想到就是求最长下降子序列。所以这题我们可以使用动态规划来实现。

以导弹依次飞来的顺序为阶段，设计状态opt[i]表示前i个导弹中拦截了导弹i可以拦截最多能拦截到的导弹的个数。

状态转移方程： 
opt[i]=max(opt[j])+1 (h[i]>=h[j],0=<j<i)   {h[i]存，第i个导弹的高度}
最大的opt[i]就是最终的解。

至于第二个问题要拦截所有导弹最少要配置的系统数。大家会被第一个问题所迷惑，大家会做最长下降子序列，然后去掉那些导弹，再求最长下降子序列，直到没有导弹。其实这是错误的不难举出反例。

6 1 7 3 2        
错解： 6 3 2/1/7   正解：6 1/7 3 2。

由于题目要求是每次打到的导弹一定要比一定要比前面打的导弹高度来的低，也就是说如果后面的导弹比前面的高那就打不到了，于是我们就能把问题抽象成求最长非下降子序列。

复杂度：时间复杂度为O（N2），空间复杂度为O（N）。

参考程序：
program missile;
const
maxn=10000;
var
a,opt:array[0..maxn] of longint;
n,anslen,anstime:longint;
procedure init;
var
x:longint;
begin
assign(input,'missile.in');
reset(input);
assign(output,'missile.out');
rewrite(output);
n:=0;
repeat
   inc(n);
   read(a[n]);
until seekeof;
end;
procedure main;
var
i,j:longint;
begin
fillchar(opt,sizeof(opt),0);
a[0]:=maxlongint;
for i:=1 to n do
   for j:=i-1 downto 0 do
    if (a[j]>=a[i]) and (opt[j]+1>opt[i]) then
     opt[i]:=opt[j]+1;
anslen:=0;
for i:=1 to n do
   if opt[i]>anslen then
    anslen:=opt[i];
fillchar(opt,sizeof(opt),0);
a[0]:=-maxlongint;
for i:=1 to n do
   for j:=i-1 downto 0 do
    if (a[j]<a[i]) and (opt[j]+1>opt[i]) then
     opt[i]:=opt[j]+1;
anstime:=0;
for i:=1 to n do
   if opt[i]>anstime then
    anstime:=opt[i];
end;
procedure print;
begin
writeln(anslen);
writeln(anstime);
close(input);
close(output);
end;
begin
init;
main;
print;
end. 
合唱队形
(chorus.pas/c/cpp)
来源：NOIP2004(提高组) 第一题

N位同学站成一排，音乐老师要请其中的(N-K)位同学出列，使得剩下的K位同学排成合唱队形。

    合唱队形是指这样的一种队形：设K位同学从左到右依次编号为1，2…，K，他们的身高分别为T1，T2，…，TK， 则他们的身高满足T1<...<Ti>Ti+1>…>TK(1<=i<=K)。

    你的任务是，已知所有N位同学的身高，计算最少需要几位同学出列，可以使得剩下的同学排成合唱队形。

【输入文件】

    输入文件chorus.in的第一行是一个整数N(2<=N<=100)，表示同学的总数。第一行有n个整数，用空格分隔，第i个整数Ti(130<=Ti<=230)是第i位同学的身高(厘米)。

【输出文件】

    输出文件chorus.out包括一行，这一行只包含一个整数，就是最少需要几位同学出列。

【样例输入】

8
186 186 150 200 160 130 197 220
【样例输出】

4
【数据规模】

对于50％的数据，保证有n<=20；

对于全部的数据，保证有n<=100。

PS：所谓的合唱队队形就是中间最高然后依次从两边递减。题目要求出队人数最少，于是我们可以抽象的想象成从左到最高的人是最长上升子序列，从最高人的人到右边是最长下降子序列。

我们来看一下复杂度：如果采用枚举中间人的话，计算最长上升或下降子序列复杂度O（n^2）

一共求N次，所以复杂度变为O（n^3）。这个复杂度对这题的数据已经可以解决，但不是最优的方法。

其实最长上升或下降子序列只要一次就可以了，因为最长最长上升或下降子序列不受中间人的影响。只要用OPT1求一次最长上升子序列，OPT2求一次最长下降子序列。这样答案就是N-max(opt1[i]+opt2[i]-1).（i是从1到n）。

这样我们就把复杂度从O（n^3）降到了O（n^2）。

参考程序：
program chorus;
const
maxn=200;
var
opt1,opt2,a:array[0..maxn] of longint;
n,ans:longint;
procedure init;
var
i:longint;
begin
assign(input, 'chorus.out');
reset(input);
assign(output, 'chorus.in');
rewrite(output);
readln(n);
for i:=1 to n do
   read(a[i]);
end;
procedure main;
var
i,j:longint;
begin
a[0]:=-maxlongint;
for i:=1 to n do
   for j:=i-1 downto 0 do
    if (a[j]<a[i]) and (opt1[j]+1>opt1[i]) then
     opt1[i]:=opt1[j]+1;
a[n+1]:=-maxlongint;
for i:=n downto 1 do
   for j:=i+1 to n+1 do
    if (a[j]<a[i]) and (opt2[j]+1>opt2[i]) then
     opt2[i]:=opt2[j]+1;
ans:=0;
for i:=1 to n do
   if opt1[i]+opt2[i]>ans then
    ans:=opt1[i]+opt2[i];
end;
procedure print;
begin
writeln(n-ans+1);
close(input);
close(output);
end;
begin
init;
main;
print;
end.
船
(ships.pas/c/cpp)
来源：《奥赛经典》（提高篇）

【问题描述】

PALMIA国家被一条河流分成南北两岸，南北两岸上各有N个村庄。北岸的每一个村庄有一个唯一的朋友在南岸，且他们的朋友村庄彼此不同。

每一对朋友村庄想要一条船来连接他们，他们向政府提出申请以获得批准。由于河面上常常有雾，政府决定禁止船只航线相交（如果相交，则很可能导致碰船）。

你的任务是编写一个程序，帮助政府官员决定批准哪些船只航线，使得不相交的航线数目最大。

【输入文件】ships.in
    输入文件由几组数据组成。每组数据的第一行有2个整数X，Y，中间有一个空格隔开，X代表PALMIA河的长度（10<=X<=6000），Y代表河的宽度（10<=Y<=100）。 第二行包含整数N，表示分别坐落在南北两岸上的村庄的数目（1<=N<=5000）。在接下来的N行中，每一行有两个非负整数C，D，由一个空格隔开，分别表示这一对朋友村庄沿河岸与PALMIA河最西边界的距离（C代表北岸的村庄，D代表南岸的村庄），不存在同岸又同位置的村庄。最后一组数据的下面仅有一行，是两个0，也被一空格隔开。

【输出文件】ships.out
      对输入文件的每一组数据，输出文件应在连续的行中表示出最大可能满足上述条件的航线的数目。

【输入样例】

30 4
7
22 4
2 6
10 3
15 12
9 8
17 17
4 2
0 0
【输出样例】

4
PS：这道题目相对来说思考难度较高，从字面意义上看不出问题的本质来，有点无法下手的感觉。这里我给大家推荐两种思考这类问题的方法。

法一：寻找规律法（我前面总结提到的第3个方法）

寻找规律自然要推几组数据，首先当然有从样例入手；





仔细观察上图：红色航线是合法的，那么他们满足什么规律呢？

                            C1    C2   C3   C4
北岸红线的端点： 4     9    15   17
南岸红线的端点： 2     8    12   17
                              D1    D2   D3   D4
不难看出无论线的斜率如何，都有这样的规律：

C1<C2<C3<C4 且 D1<D2<D3<D4
如果我们把输入数据排升序，问题变抽象为：

在一个序列（D）里找到最长的序列满足DI<DJ<Dk……且i<j<k
这样的话便是典型的最长非降子序列问题了。。。。

法二：边界条件法（我前面总结提到的第4个方法） 
边界法其实就是把数据往小了缩，显然N=1是答案是1。N=2时呢？

考虑这样一组数据：

N=2
C1    D1
C2    D2
当 C1<C2 时，如果D1>D2 那么一定会相交，反之则不会相交。

当C1
>C2时，如果D1<D2 那么一定会相交，反之则不会相交。

N=3
C1    D1
C2    D2
C3    D3
……

其实不用在推导N=3了，有兴趣的可以推导去。看N=2时就能得出：

对于任意两条航线如果满足Ci<Cj 且Di<Dj 则两条航线不相交。这样的话要想在一个序列里让所有的航线都不相交那比然满足，C1<C2<C3…Cans且D1<D2<D3…<Dans ，也就是将C排序后求出最长的满足这个条件的序列的长度就是解。

这样分析后显然是一个最长非降子序列问题。

复杂度：排序可以用O（nlogn）的算法，求最长非降子序列时间复杂度是O(n^2).
总复杂度为O（n^2）。

参考程序:
program ships;
const
maxn=5010;
type
retype=record
         C,D:longint;
        end;
var
x,y,n,ans:longint;
a:array[0..maxn] of retype;
opt:array[0..maxn] of longint;
procedure init;
var
i:longint;
begin
readln(n);
for i:=1 to n do
   read(a[i].c,a[i].d);
end;
procedure quick(L,r:longint);
var
i,j,x:longint;
y:retype;
begin
i:=L;
j:=r;
x:=a[(i+j) div 2].c;
repeat
   while a[i].c<x do
    inc(i);
   while a[j].c>x do
    dec(j);
   if i<=j then
    begin
     y:=a[i];
     a[i]:=a[j];
     a[j]:=y;
     inc(i);
     dec(j);
    end;
until i>j;
if j>L then quick(L,j);
if i<r then quick(i,r);
end;
procedure main;
var
i,j:longint;
begin
fillchar(opt,sizeof(opt),0);
quick(1,n);
for i:=1 to n do
   for j:=0 to i-1 do
    if (a[j].d<a[i].d) and (opt[j]+1>opt[i]) then
     opt[i]:=opt[j]+1;
ans:=-maxlongint;
for i:=1 to n do
   if ans<opt[i] then
    ans:=opt[i];
writeln(ans);
end;
begin
assign(input,'ships.in');
reset(input);
assign(output,'ships.out');
rewrite(output);
read(x,y);
while (x+y<>0) do
begin
   init;
   main;
   read(x,y);
end;
close(input);
close(output);
end.
背包问题
我参考了背包九讲等资料作出了如下整理。

首先说说背包问题的基本模型：

现有N个物品，每个物品的价值为V，重量为W。求用一个载重量为S的背包装这写物品，使得所装物品的总价值最高。

背包问题用贪心和搜索解，当然效果不佳，不过在我的贪心和搜索总结中会说到。显然标准的解法是动态规化，我在解决这个问题时习惯设计一维状态，还可以设计二维的，二维状态在下面会讲，现在只讨论用一维状态实现背包问题。

背包问题的分类：

（1）
小数背包:物品的重量是实数，如油，水等可以取1.67升……

（2）
整数背包：部分背包：所选物品可以是一部分。0/1背包：每个物品只能选一次，或不选。不能只选一部分

（3）
多重背包：背包不只一个。

部分背包：同小数背包。

0/1背包：这个问题是最经常出现的问题，应该熟练掌握。

    我们先看一下0/1背包的简化版：

现有N个物品，每个物品重量为W，这些物品能否使在载重量为S的背包装满（即重量和正好为S）？如过不能那能使物品重量和最重达到多少？

针对这一问题我们以物品的个数分阶段，设计一个状态opt[i]表示载重量为i的背包可否装满，显然opt[i]的基类型是boolean。

决策是什么呢？

当要装第i个物品时，如果前i-1个物品可以使载重为 k的背包装满，那么载重为k+w[i]的背包就可以装满。于是对于一个opt[j]来说，只要opt[j-w[i]]是true（表示可装满）那opt[j]就可以装满，但要注意：针对每一个物品枚举背包的载重量时如果这样正向的推导会使同一个物品用好几次，因为k+w[i]可装满那k+w[i]+w[i]就可装满，但实际上是装不满的因为物品只有一个。解决这个问题很简单，只要逆向推导就OK了。

这样划分阶段，设计状态，满足无后效性和么？

显然对与一个每一个阶段都是独立的，物品的顺序并不影响求解，因为装物品的次序不限。而对于opt[j]只考虑opt[j-w[i]]而不考虑后面的，所以满足无后效性。

有了上面的分析不难写出状态转移方程：

opt[j]:=opt[j-w[1]] {opt[j-w[i]]=true}
时间复杂度：

阶段数O(S)*状态数（O(N)）*转移代价（O(1)）=O（SN）

像装箱，采药等问题都是背包问题转变过来的。在将转移方程化为实际代码时，一共可以有3种写法，到时我会一一列举，让大家开阔思路。
装箱问题 
(pack.pas/c/cpp)
来源：NOIP2001(普及组) 第四题

【问题描述】

有一个箱子容量为V（正整数，0＜＝V＜＝20000），同时有n个物品（0＜n＜＝30＝，每个物品有一个体积（正整数）。

要求n个物品中，任取若干个装入箱内，使箱子的剩余空间为最小。

【输入文件】

    第一 行一个正整数V表示箱子的容量，第二行一个正整数N表示物品个数，接下来N行列出这N个物品各自的体积。

【输出文件】

    单独一行，表示箱子最小的剩余空间。

【输入样例】

24
6
8
3
12
7
9
7
【输出样例】

0
PS：本题是经典的0/1背包问题，并且是0/1背包的简化版，把箱子看做背包，容量看做载重量，体积看做重量，剩余空间最小也就是尽量装满背包。于是这个问题便成了：

有一个栽重量为V的背包，有N个物品，尽量多装物品，使背包尽量的重。

设计一个状态opt[i]表示重量i可否构成。

状态转移方程：opt[j]:=opt[j-w[i]] {opt[j-w[i]]=true}
最终的解就是v-x（x<=n 且opt[x]=true 且 opt[x+1..n]=false）。
1、这是用布尔数组来表示该方程代码。 
opt[0]:=true;
for i:=1 to n do
   for j:=v downto w[i] do
if opt[j-w[i]] then opt[j] :=true;
2、这是用整数数组来表示该方程代码。
opt[0]:=1;
for i:=1 to n do
   for j:=v downto w[i] do
if (opt[j-w[i]]>0)and(opt[j-w[i]]+w[i]>=opt[j]) then opt[j] :=opt[j-w[i]]+w[i];
（最后V背包能装的最大容量是opt[v]-1,因为一开始opt[0]:=1）。

3、这是另一种整数数组来表示该方程代码。
opt[0]:=0;
for i:=1 to n do
   for j:=v downto w[i] do
if (opt[j-w[i]]+w[i]>=opt[j]) then opt[j] :=opt[j-w[i]]+w[i];
（最后V背包能装的最大容量为opt[v]）。

这里大家会觉得2和3是一样的，其实如果大家跟踪的话就会发现，第2种比第3种好，为什么呢？因为第2种的替换率比第3种少，也就是说一旦数据过大的话第2种比第3种更优。
参考程序：
program pack;
const
maxv=20010;
maxn=100;
var
opt:array[0..maxv] of boolean;
w:array[0..maxn] of longint;
v,n,x:longint;
procedure init;
var
i:longint;
begin
assign(input, 'pack.in');
reset(input);
assign(output, 'pack.out');
rewrite(output);
read(v);
read(n);
for i:=1 to n do
   read(w[i]);
end;
procedure main;
var
i,j:longint;
begin
fillchar(opt,sizeof(opt),false);
opt[0]:=true;
for i:=1 to n do
   for j:=v downto w[i] do
    if opt[j-w[i]] then opt[j] :=true;
x:=v;
while not opt[x] do dec(x);
end;
procedure print;
begin
writeln(v-x);
close(input);
close(output);
end;
begin
init;
main;
print;
end. 
砝码称重 
来源：NOIP1996（提高组） 第四题

【问题描述】

    设有1g、2g、3g、5g、10g、20g的砝码各若干枚（其总重<=1000），用他们能称出的重量的种类数。

【输入文件】

a1 a2 a3 a4 a5 a6
    （表示1g砝码有a1个，2g砝码有a2个，…，20g砝码有a6个，中间有空格）。

【输出文件】

Total=N
    （N表示用这些砝码能称出的不同重量的个数，但不包括一个砝码也不用的情况）。

【输入样例】

    1 1 0 0 0 0 
【输出样例】

TOTAL=3
PS：把问题稍做一个改动，已知a1+a2+a3+a4+a5+a6个砝码的重量w[i], w[i]∈{ 1,2,3,5,10,20} 其中砝码重量可以相等，求用这些砝码可称出的不同重量的个数。

这样一改就是经典的0/1背包问题的简化版了，求解方法完全和上面说的一样，这里就不多说了，只是要注意这个题目不是求最大载重量，是统计所有的可称出的重量的个数。

参考程序：
program P4;
const
maxn=1010;
w:array[1..6] of longint=(1,2,3,5,10,20);
var
opt:array[0..maxn] of boolean;
a:array[1..6] of longint;
procedure init;
var
i:longint;
begin
for i:=1 to 6 do
   read(a[i]);
end;
procedure main;
var
i,j,k:longint;
begin
fillchar(opt,sizeof(opt),false);
opt[0]:=true;
for i:=1 to 6 do
   for j:=1 to a[i] do
    for k:=maxn downto w[i] do
     if opt[k-w[i]] then opt[k]:=true;
end;
procedure print;
var
ans,i:longint;
begin
ans:=0;
for i:=1 to maxn do
   if opt[i] then
inc(ans);
writeln(ans);
end;
begin
init;
main;
print;
end.     
积木城堡 
来源：vijos P1059
【问题描述】

    XC的儿子小XC最喜欢玩的游戏用积木垒漂亮的城堡。城堡是用一些立方体的积木垒成的，城堡的每一层是一块积木。小XC是一个比他爸爸XC还聪明的孩子，他发现垒城堡的时候，如果下面的积木比上面的积木大，那么城堡便不容易倒。所以他在垒城堡的时候总是遵循这样的规则。

    小XC想把自己垒的城堡送给幼儿园里漂亮的女孩子们，这样可以增加他的好感度。为了公平起见，他决定把送给每个女孩子一样高的城堡，这样可以避免女孩子们为了获得更漂亮的城堡而引起争执。可是他发现自己在垒城堡的时候并没有预先考虑到这一点。所以他现在要改造城堡。由于他没有多余的积木了，他灵机一动，想出了一个巧妙的改造方案。他决定从每一个城堡中挪去一些积木，使得最终每座城堡都一样高。为了使他的城堡更雄伟，他觉得应该使最后的城堡都尽可能的高。

任务：

    请你帮助小XC编一个程序，根据他垒的所有城堡的信息，决定应该移去哪些积木才能获得最佳的效果。

【输入文件】

    第一行是一个整数N(N<=100)，表示一共有几座城堡。以下N行每行是一系列非负整数，用一个空格分隔，按从下往上的顺序依次给出一座城堡中所有积木的棱长。用-1结束。一座城堡中的积木不超过100块，每块积木的棱长不超过100。

【输出文件】

    一个整数，表示最后城堡的最大可能的高度。如果找不到合适的方案，则输出0。

【输入样例】

2
2 1 –1
3 2 1 -1
【输出样例】

3
PS：把积木可搭建的最大高度看做背包的载重，每块积木的高度就是物品的重量。也就是用给定的物品装指定的包，使每个包装的物品一样多，且在符合条件的前提下尽量多。

这样就变成经典的背包问题了。

对于每一个城堡求一次，最终找到每一个城堡都可达到的最大高度即可。

参考程序：
program P1095;
const
maxhig=7000;
maxn=100;
var
n,top:longint;
opt:array[0..maxn,0..maxhig] of boolean;
a:array[0..maxn] of longint;
procedure init;
var
i:longint;
begin
readln(n);
fillchar(opt,sizeof(opt),false);
for i:=1 to n do
   opt[i,0]:=true;
end;
function can(m:longint):boolean;
var
i:longint;
begin
can:=true;
for i:=1 to n do
   if not opt[i,m] then
    exit(false);
end;
procedure main;
var
ii,m,tothig,i,j,ans:longint;
begin
for ii:=1 to n do
   begin
    top:=0;
    read(m);
    tothig:=0;
    while m>0 do
     begin
      inc(top);
      a[top]:=m;
      inc(tothig,m);
      read(m);
     end;
    for i:=1 to top do
     for j:=tothig downto 1 do
      if (j-a[i]>=0) and (opt[ii,j-a[i]]) then
       opt[ii,j]:=true;
   end;
ans:=maxhig;
while not opt[1,ans] do 
dec(ans);
while not can(ans) do
    dec(ans);
writeln(ans);
end;
begin
init;
main;
end.
阶段总结
     回顾上面的内容充分说明动态规划的本质就是递推。其实按照我的理解（动规涉及最优决策，递推是单纯的总结）背包问题的简化版更准确点算是递推而非动态规划，至于动归和递推之间的界线本来就很模糊（至少我这么认为）把它看做什么都可以，没必要咬文嚼字。

回到0/1背包的原问题上（如果你忘了就去上面看看）。

如果在不知道这个模型的情况下我们怎么做这个题呢？

这就要用到第一节提到的方法二：三要素法。

题目中明确说明对于一个物品要不就拿走要不就放下，其实题目赤裸裸的告诉我们决策就是不拿（用0表示）或拿（用1表示）。这样想都不用想就知道了决策，这也是本题的突破口。知道了决策写个搜索的程序应该是很轻松的了。

那么阶段是什么呢？

显然，给你一堆东西让你往包里塞，你当然是一个一个的那来，塞进去。那么阶段很明显就是物品的个数。

状态又是什么呢？

有的人在装东西是有个习惯（比如说我）就是先把东西分类，然后把同类的东西打个小包，最后在把小包放进去，我们可以按这个思想给物品打一些小包，也就是按照单位为1的递增的顺序准备好多小包，比如载重是6的包，可以为它准备载重是1，2，3，4，5的小包。 这样状态就可以想出来了：

设计状态opt[i，j]表示装第i个物品时载重为j的包可以装到的最大价值。

                                          opt[i-1,j]                     (j-w[i]<0，i>0)
状态转移方程：opt[i,j]= { 
max{opt[i-1,j],opt[i-1,j-w[i]]+v[i]} (j-w[i]>=0,i>0) 
(w[i]:第i个物品的重量，v[i]第i个物品的价值)
解释：要载重为j的背包空出w[i]（j-w[i]）的空间且装上第i个物品，比不装获得的价值大就装上它。

边界条件：opt[0,i]=0; (i∈{1..S})
注：这种二维的状态表示应该在下节讲，但是为了方便理解先在这里说了。

上面的方法动态规划三要素都很明显，实现也很简单。但是在我初学背包时却用另外一种一维的状态表示法。

用第一节说的思考方法五（放宽约束和增加约束）在重新思考一下这个问题：

怎么放宽约束呢？

把题目中的价值去掉，不考虑价值即最优就变成背包问题的简化版了。那简化版的求解对我们有何启示呢？

再一次增加约束：背包只能装满。

显然对于N个装满背包的方案中只要找到一个价值最大的就是问题的解。那么装不满怎么办呢？其实装不满背包，它总要达到一定的重量（X）。我们可以把这种情况看作是装满一个载重为X的小包。

总结一下上面的思维过程：

放宽约束让我们找到问题的突破口——和背包问题简化版一样，我们可以却定载重为S的背包是否可以装满。

增加约束让我们找到问题的求解方法——在装满背包的方案中选择最优的一个方案。

这样问题就解决了。

设计一个状态opt[j]表示装满载重为j的背包可获得的最大价值。对于第i个物品，只要opt[j-w[i]]可以装满且opt[j-w[i]]+v[i]比opt[j]大就装上这个物品（更新opt[j]）。

怎么使opt[j]既有是否构成又有最优的概念呢？

opt[j]只表示最优，只不过使初始条件+1，判断opt[j]是否为0，如果opt[j]=0说明j装不满。

边界条件：opt[0]=1;
状态转移方程：opt[j]=max{opt[j-w[i]]} (0<i<n,w[i]<=j<=S)
问题解： ans=max{opt[i]}-1          (0<i<=s)
时间复杂度：阶段数O(S)*状态数（O(N)）*转移代价（O(1)）=O（SN）

采药
(medic.pas/c/cpp)
来源：NOIP2005（普及组） 第三题

【问题描述】

    辰辰是个天资聪颖的孩子，他的梦想是成为世界上最伟大的医师。为此，他想拜附近最有威望的医师为师。医师为了判断他的资质，给他出了一个难题。医师把他带到一个到处都是草药的山洞里对他说：“孩子，这个山洞里有一些不同的草药，采每一株都需要一些时间，每一株也有它自身的价值。我会给你一段时间，在这段时间里，你可以采到一些草药。如果你是一个聪明的孩子，你应该可以让采到的草药的总价值最大。” 
    如果你是辰辰，你能完成这个任务吗？

【输入文件】

    输入文件medic.in的第一行有两个整数T（1 <= T <= 1000）和M（1 <= M <= 100），用一个空格隔开，T代表总共能够用来采药的时间，M代表山洞里的草药的数目。接下来的M行每行包括两个在1到100之间（包括1和100）的整数，分别表示采摘某株草药的时间和这株草药的价值。

【输出文件】

    输出文件medic.out包括一行，这一行只包含一个整数，表示在规定的时间内，可以采到的草药的最大总价值。

【输入样例】

70 3
71 100
69 1
1 2
【输出样例】

3
【数据规模】

对于30%的数据，M <= 10；

对于全部的数据，M <= 100。

PS：这是一道典型的0/1背包问题，把时间看做标准模型中的重量，把规定的时间看做载重为T的背包，这样问题和基本模型就一样了。（可以用一维和二维解决此问题）。

1、二维代码实现。
program medic;
const
maxt=1010;
maxn=110;
var
opt:array[0..maxn,0..maxt] of longint;
w,v:array[0..maxn] of longint;
t,n:longint;
procedure init;
var
i:longint;
begin
assign(input, 'medic.in');
reset(input);
assign(output, 'medic.out');
rewrite(output);
read(t,n);
for i:=1 to n do
   read(w[i],v[i]);
close(input);
end;
function max(x,y:longint):longint;
begin
if x>y then max:=x
else max:=y;
end;
procedure main;
var
i,j:longint;
begin
fillchar(opt,sizeof(opt),0);
for i:=1 to n do
   for j:=1 to t do
    if j-w[i]<0 then
     opt[i,j]:=opt[i-1,j]
    else opt[i,j]:=max(opt[i-1,j],opt[i-1,j-w[i]]+v[i]);
end;
procedure print;
begin
writeln(opt[n,t]);
close(output);
end;
begin
init;
main;
print;
end.            
2、一维代码实现。
program medic;
const
maxt=1010;
maxn=110;
var
opt:array[0..maxt] of longint;
w,v:array[0..maxn] of longint;
ans,t,n:longint;
procedure init;
var
i:longint;
begin
assign(input, 'medic.in');
reset(input);
assign(output, 'medic.out');
rewrite(output);
readln(t,n);
for i:=1 to n do
   read(w[i],v[i]);
close(input);
end;
procedure main;
var
i,j:longint;
begin
fillchar(opt,sizeof(opt),0);
opt[0]:=1;
for i:=1 to n do
   for j:=t downto w[i] do
    if (opt[j-w[i]]>0) and (opt[j-w[i]]+v[i]>opt[j]) then
     opt[j]:=opt[j-w[i]]+v[i];
ans:=-maxlongint;
for i:=1 to t do
   if opt[i]>ans then ans:=opt[i];
end;
procedure print;
begin
writeln(ans-1);
close(output);
end;
begin
init;
main;
print;
end. 
开心的金明
来源：NOIP2006（普及组） 第二题

【问题描述】

    金明今天很开心，家里购置的新房就要领钥匙了，新房里有一间他自己专用的很宽敞的房间。更让他高兴的是，妈妈昨天对他说：“你的房间需要购买哪些物品，怎么布置，你说了算，只要不超过N 元钱就行”。今天一早金明就开始做预算，但是他想买的东西太多了，肯定会超过妈妈限定的N 元。于是，他把每件物品规定了一个重要度，分为5 等：用整数1~5 表示，第5 等最重要。他还从因特网上查到了每件物品的价格（都是整数元）。他希望在不超过N 元（可以等于N 元）的前提下，使每件物品的价格与重要度的乘积的总和最大。设第j 件物品的价格为v[j]，重要度为w[j]，共选中了k 件物品，编号依次为j1...jk，则所求的总和为：v[j1]*w[j1]+..+v[jk]*w[jk]请你帮助金明设计一个满足要求的购物单.
【输入文件】

    输入的第1 行，为两个正整数，用一个空格隔开： N m 
（其中N（<30000）表示总钱数，m(<25)为希望购买物品的个数。） 
从第2 行到第m+1 行，第j 行给出了编号为j-1的物品的基本数据，每行有2 个非负整数 v p 
（其中v 表示该物品的价格（v≤10000），p 表示该物品的重要度（1~5））

【输出文件】

    输出只有一个正整数，为不超过总钱数的物品的价格与重要度乘积的总和的 最大值（<100000000）

【输入样例】

1000 5
800 2
400 5
300 5
400 3
200 2
【输出样例】

3900
PS：这仍然是一到典型的0/1背包，只不过注意这个问题中的价值对应背包模型中的重量，这个问题中的价值和重要度的成绩是背包模型中的价值。

具体实现同背包模型一样，这里不多说了。

参考程序：
program p2;
const
maxn=30010;
maxm=30;
var
opt:array[0..maxn] of longint;
v,p:array[0..maxm] of longint;
n,m:longint;
procedure init;
var
i:longint;
begin
readln(n,m);
for i:=1 to m do
   readln(v[i],p[i]);
end;
procedure main;
var
i,j:longint;
begin
fillchar(opt,sizeof(opt),0);
opt[0]:=1;
for i:=1 to m do
   begin
    for j:=n downto 1 do      
     begin
      if (j-v[i]>=0) and (opt[j-v[i]]>0) and (opt[j-v[i]]+v[i]*p[i]>opt[j])then
       opt[j]:=opt[j-v[i]]+v[i]*p[i];
     end;
   end;
end;
procedure print;
var
i,ans:longint;
begin
ans:=0;
for i:=1 to n do
   if opt[i]>ans then
    ans:=opt[i];
writeln(ans-1);
end;
begin
init;
main;
print;
end.
金明的预算方案
(budget.pas/c/cpp)
来源：NOIP2006 第二题

【问题描述】

金明今天很开心，家里购置的新房就要领钥匙了，新房里有一间金明自己专用的很宽敞的房间。更让他高兴的是，妈妈昨天对他说：“你的房间需要购买哪些物品，怎么布置，你说了算，只要不超过N元钱就行”。今天一早，金明就开始做预算了，他把想买的物品分为两类：主件与附件，附件是从属于某个主件的，下表就是一些主件与附件的例子：

主件
附件

电脑
打印机，扫描仪

书柜
图书

书桌
台灯，文具

工作椅
无

如果要买归类为附件的物品，必须先买该附件所属的主件。每个主件可以有0个、1个或2个附件。附件不再有从属于自己的附件。金明想买的东西很多，肯定会超过妈妈限定的N元。于是，他把每件物品规定了一个重要度，分为5等：用整数1~5表示，第5等最重要。他还从因特网上查到了每件物品的价格（都是10元的整数倍）。他希望在不超过N元（可以等于N元）的前提下，使每件物品的价格与重要度的乘积的总和最大。

设第j件物品的价格为v[j]，重要度为w[j]，共选中了k件物品，编号依次为j1，j2，……，jk，则所求的总和为：

v[j1]*w[j1]+v[j2]*w[j2]+ …+v[jk]*w[jk]。（其中*为乘号）

请你帮助金明设计一个满足要求的购物单。

【输入文件】

输入文件budget.in 的第1行，为两个正整数，用一个空格隔开：

N m
（其中N（<32000）表示总钱数，m（<60）为希望购买物品的个数。）

从第2行到第m+1行，第j行给出了编号为j-1的物品的基本数据，每行有3个非负整数： v p q
（其中v表示该物品的价格（v<10000），p表示该物品的重要度（1~5），q表示该物品是主件还是附件。如果q=0，表示该物品为主件，如果q>0，表示该物品为附件，q是所属主件的编号）

【输出文件】

输出文件budget.out只有一个正整数，为不超过总钱数的物品的价格与重要度乘积的总和的最大值（<200000）。

【输入样例】

1000 5
800 2 0
400 5 1
300 5 1
400 3 0
500 2 0
【输出样例】

2200
PS：这道题是提高组的就明显比普及组的拓展性。这里每个主件都可以用0，1，2个附件。

也就是说，我们可以写成这样的转移方程：(用a[i].v数组表示价格，用a[i].w数组表示重要度*价格)。

Opt[j]:=max(opt[j],opt[j-a[i].v[0]]+a[i].w[0]]);
Opt[j]:=max(opt[j],opt[j-a[i].v[0]-a[i].v[1]]+a[i].w[0]+a[i].w[1]);
Opt[j]:=max(opt[j],opt[j-a[i].v[0]-a[i].v[2]]+a[i].w[0]+a[i].w[2]);
Opt[j]:=max(opt[j],opt[j-a[i].v[0]-a[i].v[1]-a[i].v[2]]+a[i].w[0]+a[i].w[1]+a[i].w[2]);
在这里我们可以将价格做个小小的优化由于价格是10元的整倍数，所以我们可以将读进来的价格div10，输出在*10，可以有效地节约空间。（注意处理，由于一开始读进来没处理好，白白浪费10个小时的调试时间- -|||）

参考程序：
type

sz=record


x:longint;


v,w:array[0..60] of longint;

    end;
var

a:array[0..60] of sz;

i,j,k,n,m,v,p,q,max:longint;

f:array[0..10000] of longint;
begin

readln(n,m); n:=n div 10; j:=0;

for i:=1 to m do begin


readln(v,p,q);
                v:=v div 10; p:=p*v;


if q=0 then begin

        
a[i].v[0]:=v;



a[i].w[0]:=p;



inc(k);


end

        else


begin
            inc(a[q].x);



a[q].v[a[q].x]:=v;



a[q].w[a[q].x]:=p;


end;

end;
        for i:=1 to k do


for j:=n downto 0 do begin



if (j-a[i].v[0]>=0)and(f[j-a[i].v[0]]+a[i].w[0]>f[j]) then f[j]:=f[j-a[i].v[0]]+a[i].w[0];
                        if (j-a[i].v[0]-a[i].v[1]>=0)and(f[j-a[i].v[0]-a[i].v[1]]+a[i].w[0]+a[i].w[1]>f[j]) then



f[j]:=f[j-a[i].v[0]-a[i].v[1]]+a[i].w[0]+a[i].w[1];



if (j-a[i].v[0]-a[i].v[2]>=0)and




(f[j-a[i].v[0]-a[i].v[2]]+a[i].w[0]+a[i].w[2]>f[j]) then



f[j]:=f[j-a[i].v[0]-a[i].v[2]]+a[i].w[0]+a[i].w[2];



if (j-a[i].v[0]-a[i].v[1]-a[i].v[2]>=0)and




(f[j-a[i].v[0]-a[i].v[1]-a[i].v[2]]+a[i].w[0]+a[i].w[1]+a[i].w[2]>f[j]) then



f[j]:=f[j-a[i].v[0]-a[i].v[1]-a[i].v[2]]+a[i].w[0]+a[i].w[1]+a[i].w[2];



end;

max:=0;

for i:=1 to n do if f[i]>max then max:=f[i];

writeln(max*10);
end.
背包问题方案的求法
和大多数DP问题的方案的求法一样，增加一个数组path和状态维数相同用来记录当前状态的决策就OK了。

输出方案时候通过当前决策推出上一决策，这一连穿的决策序列就是要求的方案。

下面看这样一个数据：

载重：6 物品个数：3
        重量      价值

物品1： 3         10 
物品2： 2         2
物品3： 1         9
一维状态求解过程：

i=1 :   (枚举物品)
opt[0..6]= 1 0 0 11 0 0 0
path[0..6]=0 0 0 1   0 0 0 {记录最后装入包中的物品的编号}
i=2
opt[0..6]=1 0 3 11 0 13 0
path[0..6]=0 0 2 1 0 2   0
i=3
opt[0..6]=1 10 3 12 20 13 22
path[0..6]=0 3 2 3   3   2   3
二维状态求解过程： （略）

可以看到一维状态的最优解是正确的，但细心分析发现一个惊人的问题：

方案不对！！

什么最优解正确而方案不正确呢？

因为在解i=3时opt[6]用到的方案数应该是9+2+10=21。显然这个方案是真确的，所以最优解正确。但是求解完opt[6]后，接着求解opt[3]却把原来的opt[3]=10改成了opt[3]=2+9=11这样，在整个求解过程结束后最后的方案opt[6]=9+2+10就变成了opt[6]=9+2+2+9也就是说1，2两个物品装了两次。

这也正是我要说的下面的问题；

背包问题一维状态于二维状态的优劣：

显然，一维状态的维数少空间复杂度低。甚至在一些问题上可以减轻思考负担。既然这样是不是我们就应该屏弃二维状态解法呢？

由于一维状态在求解方案是存在错误，所以二维状态还是很有用啊。当然有些问题虽然也是在求解方案但要求方案唯一这样就又可以用一维状态了。

Money Systems 
(money.pas/c/cpp)
来源：USACO 2.3
【问题描述】

    母牛们不但创建了他们自己的政府而且选择了建立了自己的货币系统。[In their own rebellious way],，他们对货币的数值感到好奇。 
传统地，一个货币系统是由1,5,10,20 或 25,50, 和 100的单位面值组成的。 
母牛想知道有多少种不同的方法来用货币系统中的货币来构造一个确定的数值。 
举例来说, 使用一个货币系统 {1,2,5,10,...}产生 18单位面值的一些可能的方法是:18x1, 9x2, 8x2+2x1, 3x5+2+1,等等其它。写一个程序来计算有多少种方法用给定的货币系统来构造一定数量的面值。保证总数将会适合long long (C/C++) 和 Int64 (Free Pascal)。 
【输入文件】 
货币系统中货币的种类数目是 V (1<= V<=25)。要构造的数量钱是 N (1<= N<=10,000)。 
第 1 行: 二整数， V 和 N 
第 2 行： 可用的货币 V 个整数。

【输出文件】

单独的一行包含那个可能的构造的方案数。 
【输入样例】

3 10
1 2 5
【输出样例】

10
PS：把钱面值，把要构造的前看做载重为N的背包，这个问题便是0/1背包的简化版了，但这个问题和传统模型有所差异，不是判断N是否可构成，而是求构成N的方案，而且这里的面值是可以重复利用的（你可以看做是物品有无限多）。

对与第一个问题，只要把原来BOOLEAN型的状态改为INT64，在递推过程中累加方案数即可。

对于第二个问题，基本模型中为了避免重复在内重循环枚举背包载重时采用倒循环，现在只要正向循环就OK了。

还有一种理解就是opt[n]=opt[n-1]+opt[n-2]….
1、opt[0]:=1;
for i:=1 to v do
   for j:=a[i] to n do
     inc(opt[j],opt[j-a[i]]);
2、opt[0]:=1;
   for i:=1 to n do begin
   x:=0; 
for j:=1 to v do 
       inc(x,opt[i-a[j]]);
     opt[i]:=x;
end;
参考程序：
program money;
const
maxv=100;
maxn=10010;
var
a:array[0..maxv] of longint;
opt:array[0..maxn] of int64;
v,n:longint;
procedure init;
var
i:longint;
begin
assign(input, 'money.in');
reset(input);
assign(output, 'money.out');
rewrite(output);
read(v,n);
for i:= 1 to v do
read(a[i]);
close(input);
end;
procedure main;
var
i,j:longint;
begin
fillchar(opt,sizeof(opt),0);
opt[0]:=1;
for i:=1 to v do
   for j:=a[i] to n do
     inc(opt[j],opt[j-a[i]]);
end;
procedure print;
begin
writeln(opt[n]);
close(output);
end;
begin
init;
main;
print;
end.
新年趣事之打牌 
来源： vijos P1071
【问题描述】

    过年的时候，大人们最喜欢的活动，就是打牌了。xiaomengxian不会打牌，只好坐在一边看着。

　　这天，正当一群人打牌打得起劲的时候，突然有人喊道：“这副牌少了几张！”众人一数，果然是少了。于是这副牌的主人得意地说：“这是一幅特制的牌，我知道整副牌每一张的重量。只要我们称一下剩下的牌的总重量，就能知道少了哪些牌了。”大家都觉得这个办法不错，于是称出剩下的牌的总重量，开始计算少了哪些牌。由于数据量比较大，过了不久，大家都算得头晕了。

　　这时，xiaomengxian大声说：“你们看我的吧！”于是他拿出笔记本电脑，编出了一个程序，很快就把缺少的牌找了出来。

　　如果是你遇到了这样的情况呢？你能办到同样的事情吗？

【输入文件】

    第一行一个整数TotalW，表示剩下的牌的总重量。

　　第二行一个整数N（1<N<=100），表示这副牌有多少张。

　　接下来N行，每行一个整数Wi（1<=Wi<=1000），表示每一张牌的重量。

【输出文件】

   如果无解，则输出“0”；如果有多解，则输出“-1”；否则，按照升序输出丢失的牌的编号，相邻两个数之间用一个空格隔开。 
【输入样例】

270
4
100
110
170
200
【输出样例】

2 4
PS：因为这个问题要求多方案时输出-1，也就是说要输出的方案是唯一的，这时你就不需要担心一维状态的正确性了，可以放心的用一维求解，但要注意只有当前状态没有方案是才记录当前的方案，否则会把正确方案替换了。

参考程序：
program P1071;
const
maxw=100010;
maxn=110;
var
path,opt:array[0..maxw] of int64;
w:array[0..maxn] of longint;
ans:array[0..maxn] of boolean;
n,total:longint;
procedure init;
var
i:longint;
begin
read(total);
read(n);
for i:=1 to n do
   read(w[i]);
end;
procedure main;
var
i,j:longint;
begin
fillchar(opt,sizeof(opt),0);
fillchar(ans,sizeof(ans),true);
opt[0]:=1;
for i:=1 to n do
   for j:=total downto w[i] do
    if opt[j-w[i]]>0 then
     begin
      if opt[j]=0 then
       path[j]:=i;                      {只有当前状态没求过才记录方案}
      inc(opt[j],opt[j-w[i]]);
     end;
if opt[total]=0 then
   begin
    writeln('0');
    halt;
   end;
if opt[total]>1 then
   begin
    writeln('-1');
    halt;
   end;
i:=total;
while i>0 do
   begin
    ans[path[i]]:=false;
   i:=i-w[path[i]];
   end;
end;
procedure print;
var
i:longint;
begin
for i:=1 to n do
   if ans[i] then write(i,' ');
end;
begin
init;
main;
print;
end.
挖地雷问题 
(P3.pas/c/cpp)
来源：NOIP1996（提高组）第三题（有改动）

【问题描述】

    在一个地图上有N个地窖（N<=20），每个地窖中埋有一定数量的地雷。同时，给出地窖之间的连接路径。

当地窖及其连接的数据给出之后，某人可以从任一处开始挖地雷，然后可以沿着指出的连接往下挖（仅能选择一条路径），当无连接时挖地雷工作结束。设计一个挖地雷的方案，使某人能挖到最多的地雷。

【输入文件】

    N：                       （表示地窖的个数）

　　Ｗ1，W2，W3，……WN （表示每个地窖中埋藏的地雷数量）

     A12…………… .     A1N      
      A23…………..A2N        
             ……..
              AN-1 AN
【输出文件】

    K1--K2--……….KV             （挖地雷的顺序）

    MAX                           （挖地雷的数量）

【输入样例】

5
10，8，4，7，6
1 1 1 0
   0 0 0
       1 1 
          1
【输出样例】

5(4(3 (1 
max=27
【Hint】

题目中的路径是有向的且无环路（这是我做的改动原题中没有要求）。

PS：因为图是有向的，所以以与该顶点相连的点在往下走的路线中不包括该点。也就是说图是一个AOV网（有向无环图）。

既然满足最优化原理，且无后效性，我们就可以用动态规划解了。

这个问题的阶段就是拓扑序列，但由于输入是倒三角形，所以我们没必要求拓扑序列，只要从N到着求解就可以了。

设计状态opt[i]表示以i点出发可以挖到最多的雷的个数。

状态转移方程：opt[i]=max{opt[j]}+w[i]   (g[i,j]=1)
(g存图，w[i]存第i个地窖中的雷的个数)。

时间复杂度：

状态数O(n)*转移代价O(n)=O(n2)
这个题目还要求出路径，可以用一个辅助数组path来记录，path[i]表示从第i个出发走到的下一个点的编号。求解完只要按path记录的路径输出即可。
参考程序：
program P3;
const
maxn=200;
var
g:array[0..maxn,0..maxn] of longint;
n,ans:longint;
w,opt,path:array[0..maxn] of longint;
procedure init;
var
i,j:longint;
begin
assign(input, 'P3.in');
reset(input);
assign(output, 'P3.out');
rewrite(output);
read(n);
fillchar(g,sizeof(g),0);
for i:=1 to n do
   read(w[i]);
for i:=1 to n do
   for j:=i+1 to n do
    read(g[i,j]);
close(input);
end;
procedure main;
var
i,j:longint;
begin
fillchar(opt,sizeof(opt),0);
fillchar(path,sizeof(path),0);
for i:=n downto 1 do
   begin
    for j:=i+1 to n do
     if (g[i,j]=1) and (opt[j]>opt[i]) then
      begin
       opt[i]:=opt[j];
       path[i]:=j;
      end;
    inc(opt[i],w[i]);
   end;
ans:=1;
for i:=2 to n do
   if opt[i]>opt[ans] then ans:=i;
end;
procedure print;
var
i:longint;
begin
write(ans);
i:=path[ans];
while i>0 do
   begin
    write('-->',i);
    i:=path[i];
   end;
writeln;
writeln('max=',opt[ans]);
close(output);
end;
begin
init;
main;
print;
end.



Party Lamps
在IOI98的节日宴会上，我们有N(10<=N<=100)盏彩色灯，他们分别从1到N被标上号码。
这些灯都连接到四个按钮：
按钮1：当按下此按钮，将改变所有的灯：本来亮着的灯就熄灭，本来是关着的灯被点亮。 
按钮2：当按下此按钮，将改变所有奇数号的灯。 
按钮3：当按下此按钮，将改变所有偶数号的灯。 
按钮4：当按下此按钮，将改变所有序号是3*K+1(K>=0)的灯。例如：1,4,7... 
一个计数器C记录按钮被按下的次数。
当宴会开始，所有的灯都亮着，此时计数器C为0。
你将得到计数器C(0<=C<=10000)上的数值和经过若干操作后所有灯的状态。写一个程序去找出所有灯最后可能的与所给出信息相符的状态，并且没有重复。
PROGRAM NAME: lamps
INPUT FORMAT
不会有灯会在输入中出现两次。
第一行: N。 
第二行: C最后显示的数值。 
第三行: 最后亮着的灯,用一个空格分开，以-1为结束。 
第四行: 最后关着的灯,用一个空格分开，以-1为结束。
SAMPLE INPUT (file lamps.in)

10
1
-1
7 -1

在这个样例中，有10盏灯，只有1个按钮被按下。最后7号灯是关着的。
OUTPUT FORMAT

每一行是所有灯可能的最后状态(没有重复)。每一行有N个字符，第1个字符表示1号灯，最后一个字符表示N号灯。0表示关闭，1表示亮着。这些行必须从小到大排列（看作是二进制数）。
如果没有可能的状态，则输出一行'IMPOSSIBLE'。
SAMPLE OUTPUT (file lamps.out)

0000000000
0101010101
0110110110

在这个样例中，有三种可能的状态：
所有灯都关着 
1,4,7,10号灯关着，2,3,5,6,8,9亮着。 
1,3,5,7,9号灯关着，2, 4, 6, 8, 10亮着。
-----------------------------------

Compiling...
Compile: OK

Executing...
   Test 1: TEST OK [0.000 secs, 208 KB]
   Test 2: TEST OK [0.000 secs, 208 KB]
   Test 3: TEST OK [0.000 secs, 212 KB]
   Test 4: TEST OK [0.000 secs, 212 KB]
   Test 5: TEST OK [0.000 secs, 208 KB]
   Test 6: TEST OK [0.000 secs, 212 KB]
   Test 7: TEST OK [0.000 secs, 212 KB]
   Test 8: TEST OK [0.000 secs, 208 KB]

All tests OK.

PS:注意模拟。
{
ID:bemaula1
PROB:lamps
LANG:PASCAL
}
program Paryt_Lamps;
var light:array[1..16]of string;
    off,on:array[1..100]of integer;
    kind:array[1..16,1..4]of integer;
    a:array[1..4]of integer;
    flag:boolean;
    r,l,m,n,c,k,p:longint;

procedure init;
var k:integer;
begin
readln(n);
readln(c);
m:=c;
r:=0;
read(k);
while k<>-1 do
begin
    inc(r);
    on[r]:=k;
    read(k);
end;
readln;
l:=0;
read(k);
while k<>-1 do
begin
    inc(l);
    off[l]:=k;
    read(k);
end;
readln;
end;

procedure dfs(t:longint);
var i,j:longint;
begin
if (t=4)or(m=0)
then
begin
    a[4]:=m mod 2;
    inc(k);
    for j:=1 to 4 do
      kind[k,j]:=a[j]
end;
if (t<4)and(m>0)
then
for i:=0 to 1 do
    begin
      a[t]:=i;
      m:=m-i;
      dfs(t+1);
      m:=m+i;
      a[t]:=0;
    end;
end;

procedure check;
var try:array[1..100]of boolean;
    i,j:longint;
begin
for i:=1 to k do
begin
    fillchar(try,sizeof(try),1);
    if kind[i,1]=1 then for j:=1 to n do try[j]:=not try[j];
    if kind[i,2]=1 then for j:=1 to n do if j mod 2=1 then try[j]:=not try[j];
    if kind[i,3]=1 then for j:=1 to n do if j mod 2=0 then try[j]:=not try[j];
    j:=0;
    if kind[i,4]=1 then
    while j*3+1<=n do
      begin
        try[j*3+1]:=not try[j*3+1];
        inc(j);
      end;
    flag:=true;
    for j:=1 to r do if not try[on[j]] then begin flag:=false; break; end;
    if flag then
    for j:=1 to l do if try[off[j]] then begin flag:=false; break; end;
    if flag then
              begin
                inc(p);
                for j:=1 to n do
                   if try[j] then light[p]:=light[p]+'1'
                             else light[p]:=light[p]+'0';
              end;
end;

end;

procedure qsort(l,r:longint);
var i,j:longint;
    mid,t:string;
begin
i:=l;j:=r;mid:=light[(l+r)div 2];
repeat
while light[i]<mid do inc(i);
while light[j]>mid do dec(j);
if i<=j then begin
                 t:=light[i];
                 light[i]:=light[j];
                 light[j]:=t;
                 inc(i);dec(j);
               end;
until i>j;
if l<j then qsort(l,j);
if i<r then qsort(i,r);
end;

procedure print;
var i:longint;
begin
if p=0 then writeln('IMPOSSIBLE');
for i:=1 to p do
writeln(light[i]);
end;


begin
assign(input,'lamps.in');reset(input);
assign(output,'lamps.out');rewrite(output);
init;
k:=0;p:=0;
dfs(1);
check;
qsort(1,p);
print;
close(input);
close(output);
end.

Longest Prefix
在生物学中，一些生物的结构是用包含其要素的大写字母序列来表示的。生物学家对于把长的序列分解成较短的（称之为元素的）序列很感兴趣。
如果一个集合 P 中的元素可以通过串联（允许重复；串联，相当于 Pascal 中的 “+” 运算符）组成一个序列 S ，那么我们认为序列 S 可以分解为 P 中的元素。并不是所有的元素都必须出现。举个例子，序列 ABABACABAAB 可以分解为下面集合中的元素：
    {A, AB, BA, CA, BBC}

序列 S 的前面 K 个字符称作 S 中长度为 K 的前缀。设计一个程序，输入一个元素集合以及一个大写字母序列，计算这个序列最长的前缀的长度。
PROGRAM NAME: prefix
INPUT FORMAT
输入数据的开头包括 1..200 个元素（长度为 1..10 ）组成的集合，用连续的以空格分开的字符串表示。字母全部是大写，数据可能不止一行。元素集合结束的标志是一个只包含一个 “.” 的行。集合中的元素没有重复。接着是大写字母序列 S ，长度为 1..200,000 ，用一行或者多行的字符串来表示，每行不超过 76 个字符。换行符并不是序列 S 的一部分。
SAMPLE INPUT (file prefix.in) 
A AB BA CA BBC

ABABACABAABC

OUTPUT FORMAT
只有一行，输出一个整数，表示 S 能够分解成 P 中元素的最长前缀的长度。
SAMPLE OUTPUT (file prefix.out)
11

------------------

Compiling...
Compile: OK

Executing...
   Test 1: TEST OK [0.000 secs, 648 KB]
   Test 2: TEST OK [0.000 secs, 648 KB]
   Test 3: TEST OK [0.011 secs, 648 KB]
   Test 4: TEST OK [0.032 secs, 648 KB]
   Test 5: TEST OK [0.119 secs, 648 KB]
   Test 6: TEST OK [0.853 secs, 648 KB]

All tests OK.

PS：这道题DP，我避免进行二分查找，我将最长的集合元素长度记录下来。然后在DP的过程中用opt布尔数组表示是否出现过，最后就可以轻松得出答案。但要注意文件的读入和处理。
{
ID:bat-kil2
PROG:prefix
LANG:PASCAL
}
var
    a:array[0..200000] of char;
    opt:array[0..200000] of boolean;
    str:array[0..200] of string;
    ch:char;
    ss:string;
    i,j,k,n,m,l,maxn,maxx:longint;
    k0,k1,k2:longint;
function check(i,j:longint):boolean;
var t:longint; st:string;
begin
    st:='';
    for t:=i to j do
    st:=st+a[t];
    for t:=1 to n do
    if str[t]=st then exit(true);
    exit(false);
end;
begin
    assign(input,'prefix.in'); assign(output,'prefix.out');
    reset(input);rewrite(output);
    i:=1;
    while ch <>'.' do begin
        read(ch);
      if ch in ['A'..'Z'] then begin str[i]:=str[i]+ch; inc(maxx); end
        else
          if (ch=' ')or(ord(ch)=10) then begin
           inc(i);
           if maxx>maxn then maxn:=maxx;
           maxx:=0;
          end;
    end;
    n:=i-1; readln;
    i:=1;
    while not eof do begin
       read(a[i]);
       if a[i] in ['A'..'Z'] then inc(i);
    end;
    m:=i-1;
    fillchar(opt,sizeof(opt),false);
    for l:=1 to maxn do
        for i:=1 to m-l+1 do begin
           j:=i+l-1;
           if check(i,j) then
           for k:=i to j do opt[k]:=true;
        end;
    for i:=1 to m do
    if not opt[i] then begin writeln(i-1); halt; end;
    writeln(m);
    close(input);close(output);
end.

奶牛家谱
Silviu Ganceanu -- 2003


农民约翰准备购买一群新奶牛。 在这个新的奶牛群中, 每一个母亲奶牛都生两小奶牛。这些奶牛间的关系可以用二叉树来表示。这些二叉树总共有N个节点(3 <= N < 200)。这些二叉树有如下性质: 
每一个节点的度是0或2。度是这个节点的孩子的数目。
树的高度等于K(1 < K < 100)。高度是从根到任何叶子的最长的路径上的节点的数目; 叶子是指没有孩子的节点。

有多少不同的家谱结构? 如果一个家谱的树结构不同于另一个的, 那么这两个家谱就是不同的。输出可能的家谱树的个数除以9901的余数。
PROGRAM NAME: nocows

INPUT FORMAT

第1行: 两个空格分开的整数, N和K。
SAMPLE INPUT (file nocows.in)

5 3

OUTPUT FORMAT

第 1 行: 一个整数，表示可能的家谱树的个数除以9901的余数。
SAMPLE OUTPUT (file nocows.out)

2

OUTPUT DETAILS:
有5个节点，高为3的两个不同的家谱：
           @                       @      
          / \                        / \
         @   @      和      @   @
        / \                             / \
       @   @                  @   @
---------------------

Compiling...
Compile: OK

Executing...
   Test 1: TEST OK [0.000 secs, 288 KB]
   Test 2: TEST OK [0.000 secs, 284 KB]
   Test 3: TEST OK [0.000 secs, 288 KB]
   Test 4: TEST OK [0.000 secs, 288 KB]
   Test 5: TEST OK [0.011 secs, 284 KB]
   Test 6: TEST OK [0.011 secs, 284 KB]
   Test 7: TEST OK [0.000 secs, 288 KB]
   Test 8: TEST OK [0.032 secs, 284 KB]
   Test 9: TEST OK [0.022 secs, 288 KB]
   Test 10: TEST OK [0.000 secs, 284 KB]
   Test 11: TEST OK [0.054 secs, 288 KB]
   Test 12: TEST OK [0.043 secs, 288 KB]

All tests OK.

PS:这是一道DP题。
设f[i，j]表示i个节点，最多j深度的树的种类。 从i个结点中取出1个结点作为根结点，还剩下i-1个结点。这i-1个结点要放在这个根结点的左右子树上。而题目要求儿子数为0或是2，如果是0的话必定是叶子结点，而根结点的儿子数为2(i>1)，这样根据乘法原理，f[p，j-1]*f[i-p-1，j-1]就表示左子树有p个结点，右子树有i-p-1个结点的方案数。再根据加法原理，只要k=1..i-2且odd(k)求一下和就可以了。
       最后f[n，k]-f[n，k-1]就是“答案”（为什么加引号？看后面）。
       这当中有两点要注意，一是当i是偶数的时候没有解应当输出0，二是最后结果f[n，k]可能小于f[n，k-1]（因为为了能储存下，每计算一次f[i，j]都把它 mod 9901）所以最后结果应加上一个9901再mod 9901

f[i,j]=∑{f[i-l-1,j-1]*f[l,j-1]} (l=1 to i-2且l,i为奇数,3<=i<=n,2<=j<=k)
而f[n,k]-f[n,k-1]即为所求。
{
ID:bat-kil2
PROG:nocows
LANG:PASCAL
}
var
    opt:array[0..100,0..200] of longint;
    i,j,k,n,l,ans:longint;
begin
    assign(input,'nocows.in'); assign(output,'nocows.out');
    reset(input); rewrite(output);
    readln(n,k);
    if (n mod 2=0) then
    begin
       writeln(0);
       close(input); close(output);
       halt;
    end;
    for i:=1 to k do
    opt[i,1]:=1;
    opt[2,3]:=1;
    for i:=3 to k do
       for j:=1 to n do
          for l:=1 to j-2 do
           opt[i,j]:=(opt[i,j]+opt[i-1,l]*opt[i-1,j-1-l])mod 9901;
    ans:=(opt[k,n]-opt[k-1,n])mod 9901;
    if ans>=0 then writeln(ans) else writeln(ans+9901);
    close(input); close(output);
end.

货币系统
母牛们不但创建了他们自己的政府而且选择了建立了自己的货币系统。
[In their own rebellious way],，他们对货币的数值感到好奇。
传统地，一个货币系统是由1,5,10,20 或 25,50, 和 100的单位面值组成的。
母牛想知道有多少种不同的方法来用货币系统中的货币来构造一个确定的数值。
举例来说, 使用一个货币系统 {1,2,5,10,...}产生 18单位面值的一些可能的方法是:18x1, 9x2, 8x2+2x1, 3x5+2+1,等等其它。
写一个程序来计算有多少种方法用给定的货币系统来构造一定数量的面值。
保证总数将会适合long long (C/C++) 和 Int64 (Free Pascal)。
PROGRAM NAME: money

INPUT FORMAT

货币系统中货币的种类数目是 V 。 (1<= V<=25)
要构造的数量钱是 N 。 (1<= N<=10,000)

第 1 行: 二整数， V 和 N 
第 2 ..V+1行： 可用的货币 V 个整数　(每行一个　每行没有其它的数)。
SAMPLE INPUT (file money.in) 
3 10
1 2 5
OUTPUT FORMAT
单独的一行包含那个可能的构造的方案数。
SAMPLE OUTPUT (file money.out)
10

------------------------

Compiling...
Compile: OK

Executing...
      Test 1: TEST OK [0.016 secs]
      Test 2: TEST OK [0.016 secs]
      Test 3: TEST OK [0.012 secs]
      Test 4: TEST OK [0.012 secs]
      Test 5: TEST OK [0.036 secs]
      Test 6: TEST OK [0.04 secs]
      Test 7: TEST OK [0.016 secs]
      Test 8: TEST OK [0.012 secs]
      Test 9: TEST OK [0.012 secs]
      Test 10: TEST OK [0.02 secs]
      Test 11: TEST OK [0.012 secs]
      Test 12: TEST OK [0.02 secs]
      Test 13: TEST OK [0.016 secs]

All tests OK.

PS：这题是方案数的DP。（具体实现请看代码）。
{
ID:bat-kil2
PROG:money
LANG:PASCAL
}
var
f:array[0..10000] of longint;
        a:array[0..25] of longint; 
        v,n:longint;
i,j,k:longint;
begin
assign(input,'money.in'); assign(output,'money.out');
reset(input); rewrite(output);
readln(v,n);
f[0]:=1;
        for i:=1 to v do begin
   read(a[i]);
   for j:=a[i] to n do
    inc(f[j],f[j-a[i]]);
end;
writeln(f[n]);
        close(input); close(output);
end.

